
Overview
JawaScript is a droid programming language specifically for use on the Budget R2 J.E.D.I. R/C Control System as outlined in this document: 

www.GrayChalet.com/BudgetR2Controller.doc

The intent is to make it VERY easy for a non-skilled programmer to program the features of their droid.  It also is intended to have enough features as to not inhibit almost all possibilities of the system.  However, there are always those individuals who want to “tweak” and “customize”.  Experienced programmers can compile their own custom code for the Budget R2 J.E.D.I. R/C Control System easily using the free Integrated Development Environment (IDE) available from Parallax Systems (www.Parallax.com ).  

JawaScript is an interpreted language.  This means that the code is typically placed into an ASCII text file and then executed from that file at run time (from a USB thumbdrive in the droid).  The term typically is used because the commands can also be entered at a command line prompt over a TTL-level RS232 interfaces.  This would allow a PC to also control a droid in lieu or in combination with the USB Flash thumbdrive.  The RS232 interface used is commonly called a Command Line Interface (CLI).  

There are four ways that JawaScript files can be invoked:

1.
JawaScript files are executed in response to a Joystick Enhanced Device Input (J.E.D.I.) command from an R/C transmitter. 

2.
JawaScript files are executed in response to a JawaScript command from an RS-232 Command Line Interface (CLI).  JawaScript commands can also be entered directly from the CLI.
3.
A call or jump from another JawaScript file. 

4.
At startup from a power-on reset, a user pressed button reset, or a JawaScript fault due to too deep of nested calls.  This will cause the file named “Startup.txt” to be executed, if found on the thumbdrive.  Note that the customization of the controller for a particular droid can be done just using JawaScript!

Since there is limited executable memory in the Propeller microcontroller, and adding memory for program execution is not really feasible, the JawaScript language has to be very basic.  User defined variables are not currently supported.   Conditional instructions don’t exist (yet).  Nesting (calling other JawaScript files to run and then return to the calling script) will be limited to 2 deep, when implemented.  But, it also has to support the many features and coordinate activities within a droid.  This is a delicate balance, and every attempt is being made to be as flexible as possible without breaking the bank.  Note that normally, JawaScripts are simply setting up outputs and assigning R/C receiver channels at the response of a JEDI command.  

This document defines the language constructs, but not the actual results of the commands.  This is because most command results are dependent upon usage in the droid.  At first, this may seem daunting, but with some experience using the CLI to a PC, things will become clear (practice makes perfect…).

Document Revision History

Revision X00: 06/20/2007 (Scott Gray) 

Initial Rough Draft sent for comments (Scott Gray)

Revision X01: 07/16/2007 (Scott Gray, et al)


Clean-up spelling and syntax mistakes! (Scott Gray)

Added LIMIT attribute to SERVO (Scott Gray… Oops, forgot!  Was supposed to be there in last revision!).

“Added” MODE attribute to many objects (Scott Gray – cleans up syntax, easier parsing, mode attribute was implicit, now made explicit)

Added mode attribute to PSI for random, other effects (Mike Velcheck)

Added COM command to send message to a slave device.  This is just a special case of DISPLAY command. (Mike Velcheck input)


Changed PLAY to require which MP# player to use, although only 1 is defined.  This cleans up syntax to be same as other objects, and allows for future growth. (Scott Gray)


Added SENTENCE as allowed attribute to PLAY command.  Was already in syntax, just not explicitly described. (Scott Gray)
· Fixed SCALE attribute description of pushing range limits. (Scott Gray)   

· Changed servo SET value to microseconds instead of degrees for consistency. (Scott Gray)
· Added some more introduction material and reference section (Lynelle Phillips input)
· Added Acronyms list (Scott Gray)


Added Glossary (Scott Gray)

Revision V1.0: 07/29/2007 (Scott Gray, et al)

· More clean-up of spelling and syntax mistakes! (Scott Gray)

· Fixed PARK example errors (Scott Gray)

· Added COM command to overview of commands page (Scott Gray)
· Added PATROL, ALARM, and MISC attributes to PLAY command (George Luck)

· Added explanation that objects are enumerated explicitly (Bob Ross)
· Added Set JEDI syntax (Scott Gray)

· Changed RANGE attribute for servos to microseconds instead of degrees since the number of degrees depends on servo type (Scott Gray)

· Added reference to FTDI RS232 cable.

References

www.GrayChalet.com/BudgetR2Controller.doc - Overview material for the controller that JawaScript runs on.

www.superdroidrobots.com/product_info/RC.htm - Information about R/C basics.
www.easyrc.com/radiosystems/index.html  -  For more definitions, including a discussion on types of servos and batteries.


www.Parallax.com – Information on Propeller microcontroller that JawaScript executes on (using Spin and assembly).  

tech.groups.yahoo.com/group/VEX_Modders  - This forum is for discussion of modifying, using, programming, and engineering input/output devices for the VEX Radio Control system.  The J.E.D.I. R/C Control System is discussed there.

www.ftdichip.com/Products/EvaluationKits/TTL-232R-3V3.htm - USB to 3.3V TTL compatible RS232 cable.  Useful for connecting devices that use JawaScript to a PC for testing, debug, etc.


Acronyms
There are many acronyms used throughout this reference manual.  Most are listed here for reference.

	AI  
	Analog Input – 0-5V Analog to Digital inputs for sensors

	ATV
	Adjustable Travel Volume – adjusts servo movement range

	BL
	Back Light (used by LCD displays)

	CLI 
	Command Line Interface – scripting input

	DI
	Digital Input – used for switches and other discrete inputs

	DOUT
	Digital Output – used for communicating to other effects

	FLD
	Front Logic Display – The two displays on lower front of dome

	FS
	Fail Safe – upon failure detection, to return all outputs to known safe state

	HCOUT
	High Current Output – used for driving effects

	HP
	Holo-Projector – The three moving lamps on dome

	JEDI
	Joystick Enhanced Device Input

	LCD
	Liquid Crystal Display – display for status, debug – not visible on droid

	LD
	Logic Display – displays on dome front and back

	MIC
	Microphone

	MP3
	Compressed media format for music, sounds

	PIR
	Passive Infra-Red (sensor)

	PSI
	Process State Indicator – The blinking circles

	R/C
	Remote Control

	RLD
	Rear Logic Display – display on back of dome


Glossary

There are many terms that may be unfamiliar used in this reference manual.  Some might be listed here for reference.

Adjustable Travel Volume - ATV allows you to preset the maximum travel of a servo to either side from its neutral position. Such settings help tailor control servo action. 

Fail Safe - This feature automatically returns servos to neutral or a preset positions in case of a malfunction, loss of R/C signal,  or interference. 

Inverse -  This means to reverse servo or motor rotation, or to change polarity of a digital output from active high for true state to active low.  
Park – Parking an R/C channel means to freeze (park) all outputs currently connected to that channel and then free up that channel for another use.  A good example is when entering a J.E.D.I. command: The outputs connected to the two channels being used for a J.E.D.I. command stroke input are frozen while the command is entered.  For example, if channel 4 is used to control dome rotation, then the dome could be put in motion by pushing the joystick over, and then releasing the joystick by pressing the J.E.D.I. command button (or toggle switch, depending upon transmitter), the dome would still rotate based on the last joystick position prior to pressing the button.  

Servo – Servos are DC motors with built-in gearing and electronic feedback control loop circuitry for positioning actuator arms (called servo arms) based on a supplied input. A servo contains an electric motor and is the "muscle" that controls droid features such as opening and closing doors, moving the Holo-projectors, etc.  There is a wide variety of servo types depending upon their intended use.   For a more detailed look at servos, including the different types and how to hook one up, see the references listed in the reference section.
Slew – Slew refers to the rate that an output can change value.  For example, slewing an LED determines how fast the LED turns on or off (how fast it fades).  Slewing a servo refers to how fast the servo can change from present position to the new inputted position (joystick position, for example).  Servos are inherently slew limited based on the servo type, but often there is a need to move more slowly to the new position to prevent damage of the controlled object.

Syntax Definitions
The following syntax definitions are used for describing the syntax of the JawaScript commands. The syntax definitions use special symbols to indicate when and how certain element features are to be used:


BOLDCAPS
Items in bold uppercase should be typed in as shown.


Bold Italics
Items in bold italics should be replaced by value or object


 ,  “  ()  
Commas, double quotes, and parentheses should be

typed in as shown.


<>

Angle bracket symbols enclose optional items. Enter the

enclosed item if desired. Do not enter the angle brackets.


(( | )) 

Double parentheses symbols enclose mutually-exclusive

items, separated by a vertical line. Enter one, and only one,

of the enclosed items. Do not enter the double parentheses or the vertical line.


{CR}

New Line symbology indicates must be end-of-line.


{space}
Space symbology or a space indicates items should be

separated by at least one space.


‘

The ‘ character is an escape character to ignore rest of line, used for

commenting JawaScript code or temporarily removing code


…

Three periods means the syntax definition continues on the next

line, the command must all be on one line. 

Note that JawaScript is case insensitive, this means that This, ThIs, tHiS, and thiS are all equivalent to the command THIS, the way it is shown in this document (had THIS been a command, of course).

Defined Variables

While user variables are not supported, several pre-defined variables are included for quite a bit more functionality using JawaScript.  These variables can be used in several commands to provide additional information as well.


LOWLIMIT
Use of this variable varies on the command or situation, but it
sets the lower bounds of an operation and can be set by the user. 


HIGHLIMIT
Use of this variable varies on the command or situation, but it
sets the higher bounds of an operation and can be set by the user. 


RANDOM
Use of this variable will generate a random integer number that
is greater than or equal to the value in variable LOWLIMIT and 

less than or equal to the value in variable HIGHLIMIT.  Can be used wherever a number is called for in a JawaScript command.  This is both powerful and very dangerous. Here’s an example of the danger: Consider using a random number for assigning a channel to some random output.  The droid handler would have a tough time figuring out what his joysticks were controlling!  Use the RANDOM variable with caution.


SERVONUM
This variable is a fixed value of the number of servos
supported.  Not used yet by any commands, but a good reference.


CHANNUM
This variable is a fixed value of the number of receiver
channels supported. Not used yet by any commands, but a good reference.


HCOUTNUM
This variable is a fixed value of the number of high current
digital outputs supported.  Not used yet by any commands, but a good reference.

DOUTNUM
This variable is a fixed value of the number of  TTL level
digital outputs supported. Not used yet by any commands, but a good reference.
Defined JawaScript Objects

JawaScript manipulates “objects” that the Budget R2 J.E.D.I. R/C Control System provides.  These objects are what make up the ability of the controller to perform droid functions.  There can be any number of each of these objects based on the JawaScript language, but the controller only has a certain number of each object.  In no particular order:

Outputs

SERVO
Standard R/C PPM Servo output

PSI

Process State Indicator

FLD

Front Logic Display

RLD

Rear Logic Display

HP

Holo-Projector

DOUT

Digital Output, TTL-level

HCOUT
High Current Digital Output

MP3

VMusic2 MP3 Player

LCD

20 by 4 character LCD display

Inputs

AI

Analog Input

DI

Digital Input

CHANNEL
Receiver Channel

PIR

Passive InfraRed sensor

MIC

Microphone input

JEDI

JEDI command input

Defined Object Attributes

Some of the objects that JawaScript controls have attributes that can be modified by JawaScript commands.   These attributes for each object are described below:

SERVO
The SERVO object has the following attributes:

o
CENTER – Center servo pulse width value, in microseconds, default is 1520 microseconds.

o
DEADBAND – Deadband servo value, in microseconds. Provides a deadband for which servo control input (such as from a receiver channel) does not make servo leave center value.  This prevents servo chatter or motor hum and controls sensitivity to slight joystick movements.  Default is 20 microseconds.
o
SLEW – Slew rate of servo, in degrees per second.  Provides for control of how fast a servo moves from one value to another.  The default is the special value of 0, which means no slew rate control.
o
RANGE – Servo travel range, in microseconds.  Prevents servos from going to far by setting a servo’s ATV.  Attribute is set using the LOWLIMIT and HIGHLIMIT defined variables.  Set them to desired low range and high range limits.  Default is 1000 microseconds for low end of range, and 2000 microseconds for top end.  This is standard full range for standard servos.  
o
SCALE – Servo input scaling multiplier, in 1/8th increments.  Allows for more or less sensitivity to joystick inputs and increases or decreases range.  For example, a scale of 8 (8 1/8th increments equals scaling of 1) provides normal 1 to 1 scaling of a servo’s input.  A value of 4 halves the sensitivity and range.  This is useful for the drive motors!  A value of 16 doubles sensitivity and range (for pushing standard servos to their limits!).
o
LIMIT – Servo travel limit switch inputs.  Prevents servos from going to far based on Digital Inputs.  Attribute is set using the LOWLIMIT and HIGHLIMIT defined variables.  Set them respectively to desired shorter pulse width DI and longer pulse width DI.  If one of the limit variables is set to 0, then there is no limit DI assigned for that servo direction.  Default is zeros (no limit DIs defined).  Active state of limit is default of true (1).  To use false (0) as active limit use the INVERSE modifier.
PSI

The PSI object has the following attributes:

o
MODE – Sets a PSI mode. Default is 1, which is normal alternating colors.

o
RATE – Color switching rates, in milliseconds. Attribute is set using the LOWLIMIT and HIGHLIMIT defined variables.  LOWLIMIT defines first color time in milliseconds, HIGHLIMIT defines second color time in milliseconds. 

o
TRANS – Transition rate from color to color, in milliseconds.  For example, the front PSI will have both red and blue on for TRANS time.  Default is 250 milliseconds based on observation.

o
SLEW – Color slew rate, in milliseconds.  The PSI colors can slew from one color to the next by fading off one color while fading in the other.  This gives a fiber optic effect to the PSIs.  The default is 0, so there is no slew.  This is screen accurate.

FLD

The FLDs have the following attributes:

o
MODE – Sets a FLD mode. Default is 1, which is normal random display.

o
SLEW – Slew rate of LED on-off transitions, in milliseconds.  Creates the fiber-optic look.  The default value is 40 milliseconds based on preference.

o
RATE – Rate of pattern transitions, in milliseconds.  This attribute’s effect depends upon pattern selected.  Not all patterns use this attribute.  See FLD documentation for patterns that are defined. Default is 80 milliseconds for the initial random pattern based on preference.

o
MESSAGE – Message to scroll on display, which is an ASCII string of 40 characters or less.  Used in conjunction with the message pattern mode of the FLD.  Default is “J.E.D.I. R/C Control FLD”.
RLD

The RLDs have the following attributes:

o
MODE – Sets a RLD mode. Default is 1, which is normal random display.

o
SLEW – Slew rate of LED on-off transitions, in milliseconds.  Creates the fiber-optic look.  The default value is 40 milliseconds based on preference.

o
RATE – Rate of pattern transitions, in milliseconds.  This attribute’s effect depends upon pattern selected.  Not all patterns use this attribute.  See FLD documentation for patterns that are defined.  Default is 100 milliseconds for the initial random pattern.

HP

The HPs have the following attributes:

o
MODE – Sets an HP mode. Default is 1, which is normal flicker display.

o
ON – Turns on HP

o
OFF – Turns off HP, default.

DOUT

The DOUTS have the following attributes:

o
ON – Turns on a Digital Output

o
OFF – Turns off a Digital Output, default.

HCOUT
The HCOUTS have the following attributes:

o
ON – Turns on a High Current Digital Output

o
OFF – Turns off a High Current Digital Output, default.

MP3

The MP3 has the following attributes:

o
MODE – Sets the MP3 player’s mode.  Default is mode 1, which only plays sound when PLAY command executed. 

o
VOL – Volume adjust, in percent.  Default is 85 for start sounds.  Gives room for Scream and Short Circuit to get louder ;).

o
OFF – Turns off currently playing sound, if any.  No default. 

LCD

The LCD has the following attributes:

o
MODE – Sets the LCD display type. Default is 1.
o
BL – BackLight, 1 or 0.  1 for on, 0 for off.

o
CLEAR – Clears LCD display, cursor at 0, 0. Only useful if LCD display type is 0.

o
XPOS – X position on a line, 0 to 19 character positions.  Places cursor at the specified position on current line. Only useful if LCD display type is 0.

o
YPOS – Y position (line) on display, 0 to 3 line positions.  Places cursor on the specified line. Only useful if LCD display type is 0.

JawaScript Commands

JawaScript is comprised of only a few commands that provide a lot of functionality.  A quick description is given here, and a detailed one of each command follows.


ASSIGN
Assigns a servo or output to a channel for control.


CALL

Executes another JawaScript file and then returns to the
current JawaScript file. NOT IMPLEMENTED IN FIRST RELEASE.

COM

This command is used to send a command string to another




Controller via RS232.


DISPLAY
Displays text, number, or variable value at current cursor position
on LCD display. 


PARK

Frees a channel for other use by taking its current assignments

and placing those servos or outputs into a “parked” state with a frozen current value of the channel.


PLAY

Plays a sound or song file (MP3 or WAV) from thumbdrive.  If a

current sound or song is playing, it will abruptly stop and the new sound or song will play.


RESET

Resets a servo, PSI, FLD, RLD, HP, or output to its

initial state. Resets a CHANNEL to its initial assignment and value.


RUN

Executes, or “Runs” another JawaScript file.  Does not return 

to the current JawaScript file.


SET

Sets a servo, channel, PSI, FLD, RLD, HP, MP3, J.E.D.I. command




stroke, output or one of  its attributes to a new value. 


UNPARK
Re-assigns a channel’s assignments to its last set of 

assignments, if any.  If there are none, the channel is free.


WAIT

Waits a period of time, or until an event occurs such as a
sensor input state.  Also can wait until next JEDI command is entered that is not a pre-defined sound command. 

ASSIGN

Syntax:

ASSIGN (( SERVO | DOUT | HCOUT )) OutputNum, ((ChannelNum | OFF))…

 <, INVERSE>

Assigns an output to a channel input.  If the channel number is less than or equal to CHANNUM (the number of receiver channels supported by the controller H/W), then that receiver channel will now control that output directly.  If the channel number is greater than CHANNUM, then the output is assigned to a virtual channel (a channel that must be explicitly set and not controlled by R/C receiver).  The total number of channels supported is 16.  The INVERSE modifier will invert the channel input to the output.  This allows for reversing a motor or servo direction from what the receiver channel outputs.

Note the using a SET command to set an output that is assigned to a receiver channel is not a good idea.  The value will last only at most about 1/50th of a second until the next received channel command.  Virtual channels will be evaluated only when the value is changed.  So in theory, an output assigned to a virtual channel could also use a SET command, but it would get very confusing.  Don’t do it!

Examples:
ASSIGN SERVO 7, 3


‘Assign servo 7 to receiver channel 3 

ASSIGN SERVO 8, 12


‘Assign servo 8 to virtual channel 12 

ASSIGN SERVO 2, 2, INVERSE
‘Assign servo 2 to receiver channel 2, but inverse commands, useful for controlling droids with motors


or servos that want to run  “backwards” from the receiver input .

ASSIGN HCOUT 1, 5
‘Assign High Current Output to VEX channel 5 which is a digital channel.  Note that setting the HCOUT with a SET command has only a transient effect until next received channel input (about 1/50th of a second).

ASSIGN DOUT 2, OFF

‘Un-assign DOUT 2, reverts to default value

CALL (Defined, but initial release will not support)

Syntax:

CALL “JawaScriptFile.txt”

Executes another JawaScript file named JawaScriptFile.txt and then returns to the
current JawaScript file.  Only 2 levels of calls are expected to be supported.  If another level is called, the JawaScript file and all its calling scripts are aborted, and the Startup.txt JawaScript is then run to clean up the mess.  

This command will not be in the initial release of JawaScript as it is VERY difficult to implement correctly, and will require time to test out.

Example:
CALL “YOURMOM.TXT”
‘Calls JawaScript file YOURMON

RUN “FOREST.TXT”

‘After returning, runs FOREST

DISPLAY “Here!”

‘Never executes this as last command was RUN 

COM

Syntax:

COM ComNum ((,“ASCII_Text” | ,DefinedVariable))

This command is used to send a command string to another controller by RS232. The longest the text string can be is 20 characters.  The string will be sent without a delimiter character at the end.  So to send a carriage return ({CR}, a.k.a. enter key), for example, requires setting up a defined variable with the decimal value 13 and sending that variable.  There is no support for hardware handshaking, so only send commands at a rate the receiving controller can accept (use WAIT command if needed).
Sending longer strings can be done by multiple COM commands.  

Examples:

COM 1, “TURN OFF WATER”

‘Sends string shown to slave device 1.

SET LOWLIMIT 13

COM 1, LOWLIMIT


‘Sends Carriage Return to slave device 1.

DISPLAY

Syntax:

DISPLAY LCDNum ((,“ASCII_Text” | ,DefinedVariable | LCDAttribute ))

This command is used to display characters on the 20 X 4 character LCD display.  

The longest the text string can be is 20 characters, but depends upon cursor position.  See SET command for examples of placing cursor on display.  Any of the defined variables can be displayed.  Note that a {CR} is NOT done automatically.  If you wish to 

go to start of new line then you must use SET LCD XPOS, and SET LCD YPOS or have exactly 20 characters so that the next character will wrap.

The LCD display type is set using the “SET LCD 1 MODE #” command.  A number representing a predefined display format is given.  For example, 1 is the JEDI cursor screen.  It automatically is displayed when a JEDI command is being entered, but you can make the screen remain by setting the display type as 1 using SET LCD 1 MODE 1.  But, then using DISPLAY only momentarily will display something (probably in the wrong place), until the next display refesh dependent upon the display type setting.  Bottom line, don’t use this command unless the LCD is set as the user-defined type.   

Example:

SET LCD 1 MODE 0


‘Set LCD 1 display type to 0, which is user defined.

SET LCD 1 CLEAR


‘Clear the display

DISPLAY 1, “This is a test: ”

‘Displays text shown.

DISPLAY 1, LOWLIMIT

‘Displays LOWLIMIT value on same line as above.

DISPLAY 1, SERVONUM

‘Displays supported number of servos
PARK

Syntax:

PARK ChannelNum 

Frees a channel for other use by taking its current assignments and placing those servos and/or outputs into a “parked” state with a frozen current value of the channel. Effectively, this is like re-assigning all outputs using the channel to another virtual channel with a fixed value.  Only 1 level of parking is supported.  If more than PARK command is performed on a channel, then on a restore using the UNPARK JawaScript command, all outputs that were parked will now go back on that channel.  Watch out for JawaScripts that are CALLed!

Example:
PARK 3             


‘Park the current channel 3 assignments

ASSIGN SERVO 2, 3, INVERSE 
‘Channel 3 now controls Servo 2, inverse

WAIT JEDI         
    

‘Wait for new J.E.D.I. or Null command


ASSIGN SERVO 2, OFF

‘ Place servo 2 back to default

SET SERVO 2, 1520
     

‘Set servo 2 to 0 degrees 

UNPARK 3                  

‘Restore channel 3 output assignments

PLAY

Syntax:

PLAY MP3PlayerNum, (( “SoundFile.(( MP3 | WAV))” | PROCESS | WHISTLE |…

          HUM | RAZZ | SENTENCE | PATROL | ALARM | MISC))

Plays a sound or song file (MP3 or WAV) from thumbdrive.  If a current sound or song is playing, it will abruptly stop and the new sound or song will play.  

There are also some special attributes for this command that play a random MP3 sound file from a group of sounds categorized by type.  These sounds are all placed in individual file folders with the same name as the attribute.  There must be a file named “NUMBER.TXT” in each folder that has a number, in ASCII, which is the number of sounds in that folder.  The sounds must be named the same as the attribute, followed by a number from 1 to N, where N is the number of sounds.  For example, a folder named PROCESS that has 3 sound files named PROCESS1.MP3, PROCESS2.MP3, PROCESS3.MP3 and a NUMBER.TXT file that has only the single ASCII digit “3” in it.  This allows easy customization and expansion of the number of random sounds of each type.

PLAY Special Attributes for playing random sounds from a type:


PROCESS : Typical R2 sounds while computing

WHISTLE : Typical R2 whistle sounds

HUM : Typical hums, rattle sounds from R2

RAZZ
: R2 razzing sounds

SENTENCE : Longer combinations of sounds from R2

PATROL : Like when R2 is going off on his own.  Examples: in the desert seeking out Obi-wan, swamp getting out, etc.  Just a "happy" cruise mode sound set.

· ALARM  : When R2 goes into an alarm situation.  Example:  detects intruders, senses danger (think "sand people" on ANH).

· MISC : For miscellaneous random sounds....

Example:
PLAY 1,”/MUSIC/StarwarsTheme.MP3”

WAIT MP3IDLE

PLAY 1,”Scream.wav”

WAIT MP3IDLE

PLAY PROCESS
‘Plays a random PROCESS#.MP3 sound from PROCESS folder

RESET

Syntax:

RESET (( SERVO | DOUT | HCOUT | PSI | FLD | RLD | HP | CHANNEL)) …

(( OutputNum | ChannelNum ))

This command resets a servo, PSI, FLD, RLD, HP, or output to its default state.  It also will reset a CHANNEL to its initial assignment and value.

Examples:
RESET SERVO 7


‘Resets servo 7 to center position 

RESET CHANNEL 1


‘Resets Channel 1 assignment to what was assigned






‘in Startup.txt

RUN

Syntax:

RUN “JawaScriptFile.txt”

Executes another JawaScript file named JawaScriptFile.txt but does not return to the
current JawaScript file.  Note that if a JawaScript file is CALLed by another JawaScript file, and it then RUNs another JawaScript file, when that file completes, execution will continue with the original JawaScript file that made the call.  Note that JawaScript files can be strung together for as much as you wish using RUN commands.  Also note that the RUN command better be the last command in a JawaScript file that uses it.

Examples:
RUN “Forest.txt”
‘Runs JawaScript named Forest.txt

SET

Syntax:

SET ((SERVO | DOUT | HCOUT | PSI | FLD | RLD | HP … 
| CHANNEL | MP3 | LCD | JEDI)) …

(( OutputNum | ChannelNum | JEDINum ))…
((, IntegerValue | OutputAttribute |, JEDIStroke))… 

<(( IntegerValue | “ASCIIMessage” | “JawaScriptFile”)) >

Sets a servo, channel, PSI, FLD, RLD, HP, MP3, JEDI stroke, output, or one of an output’s attributes, to a new value.   

Examples:
SET PSI 1 MODE 1
      
‘Set Front PSI to normal constant blink mode 

SET PSI 2 MODE 1
      
‘Set Rear PSI to normal constant blink mode 

SET LOWLIMIT 3000

‘Set defined variable to value for use later

SET HIGHLIMIT 1000

‘Set defined variable to value for use later

SET PSI 1 RATE
      
‘Set Front PSI initial times (msec)  

SET PSI 2 RATE
      
‘Set Rear PSI initial times same as front (msec)  

SET PSI 1 TRANS 250          ‘Set PSI transition time (msec)

SET FLD 1 MODE 1
          
‘Set Top FLD to random pattern

SET FLD 1 SLEW 500         
‘Set Top FLD to slew rate to 500 milliseconds

SET FLD 2 RATE 150         
‘Set Bottom FLD rate to 150 milliseconds

SET FLD 2 MESSAGE “THIS IS A TEST MESSAGE”

SET FLD 2, 5

          
‘Set Bottom FLD to pattern 5 (scroll message)

SET RLD 1 MODE 1

‘Set RLD to random pattern

SET RLD RATE 100

‘Set pattern rate on RLD 1 to 100 milliseconds

SET HP 1 MODE 1

‘Set Holo-Projector 1 to pattern 1 (flicker)

SET HP 1 ON


‘Set Holo-Projector 1 on

SET HP 2 OFF


‘Set Holo-Projector 2 off

SET CHANNEL 2, 1520         ‘Set Initial channel 2 center value

SET DOUT 2 ON 

‘Set Digital output 2 on

SET HCOUT 3 OFF

‘Take a leaping guess

SET CHANNEL 3, 1520
‘Set channel 3 to center value of 1520 microseconds

SET MP3 1 VOL 100

‘Full volume on MP3, baby!

SET LCD 1 BL 1

‘Set LCD 1backlighting on

SET LCD 1, 0


‘Set LCD 1 display type to 0, which is user defined.

SET LCD 1 CLEAR

‘Clear the display

SET LCD 1 XPOS 5

‘Put cursor at character 5 on line 0 (because of clear)

SET LCD 1 YPOS 2

‘Change to line 2, cursor is at character 5 on line 2

DISPLAY 1, “This is a test.”  ‘Prints message, left justified since start at character 5

SET SERVO 2 CENTER 1520

‘Set servo 2 center to 1520 microseconds  

SET SERVO 2 DEADBAND 20
‘Set servo 2 deadband to 20 microseconds

SET LOWLIMIT 3


‘Set defined variable for short pulse limit switch DI

SET HIGHLIMIT 4


‘Set defined variable for long pulse limit switch DI

SET SERVO 2 LIMIT INVERSE
‘Set servo 2 limit switches as DIs 3 and 4, active low

One special attribute that the SET command can do is link a J.E.D.I. command stroke to a JawaScript file to execute.  The command stroke is defined by how a joystick moves through 9 “regions” of the joystick’s range of motion.  A J.E.D.I. command stroke that can be defined always starts away from center stick, and ends away from center stick.  It also passes through no more than 8 regions.  The regions are defined as follows (as a joystick is viewed from above, with center stick being region 5):

	1
	2
	3

	4
	5
	6

	7
	8
	9


A stroke from upper left to upper right is pre-defined as the “start over”, or reset, J.E.D.I. command, and re-initializes the controller and then runs the Startup.txt file.  This could have been defined using the following JawaScript command (if not already pre-defined):

SET JEDI 2, 123 “Startup.txt”
‘Set JEDI stroke 123 to run JawaScript Startup.txt

A J.E.D.I. stroke definition for controlling a periscope might be a stroke from middle bottom to middle top:
SET JEDI 2, 852 “Periscope.txt”

Note that most pre-defined J.E.D.I. strokes use the first J.E.D.I. button (upper left button on a VEX radio), although there are a few pre-defined strokes on the second J.E.D.I. button (lower left button) as well.  If a stroke is pre-defined by the controller, it will NOT be overwritten by a JawaScript file using the “SET JEDI…” command.  Only un-defined J.E.D.I. strokes can be set to run a JawaScript file.
UNPARK

Syntax:

UNPARK ChannelNum
This command re-assigns a channel’s assignments to its last set of assignments, if any.  If there are none, the channel is not controlling anything (it is free to use).  This un-does what a PARK command did to a channel.  See the PARK command for better description.

Examples:
See PARK command for examples.

WAIT

Syntax:

WAIT (( TimeValue | JEDI | PIR | MIC | AI | DI | CHANNEL | MP3IDLE )) …

<(( TRUE | FALSE ))>

Waits a period of time in milliseconds,  or until an event occurs such as a sensor input state.  Also can wait until next JEDI command is entered.  The AI, MIC, PIR, and CHANNEL inputs use the LOWLIMIT and HIGHLIMIT variables to define when an event occurs.  Each is explained below in the examples.

Examples:
WAIT 150
‘Waits 150 milliseconds before executing next JawaScript command.

WAIT JEDI
‘This waits for the next non-sound related JEDI command to be entered.  ‘This is useful when a feature has been set up by this script, and when the ‘user enters a new JEDI command, the script then finishes by restoring ‘channel assignments, turning off the feature, etc. 

WAIT PIR
‘This will wait until LOWLIMIT events are detected by the PIR Sensor.  By ‘changing the value in the defined variable LOWLIMIT the sensitivity of the ‘PIR can be controlled.  A good value to start with seems to be 5.

WAIT MIC
‘Waits for sound threshold to be greater than value in LOWLIMIT.  By ‘changing the value in the defined variable LOWLIMIT the loudness detected by the MIC can be controlled.
WAIT AI 3
‘Waits for analog input 3 to be in range from LOWLIMIT to HIGHLIMIT.  By ‘changing the value in the defined variables LOWLIMIT and HIGHLIMIT, the analog input detected can be set.

WAIT DI 4 TRUE
‘Waits for Digital input 4 to be a TTL-level high (> approx 2.4V).

WAIT DI 2 FALSE
‘Waits for Digital input 2 to be a TTL-level low (< approx 0.8V).

WAIT CHANNEL 3
‘Waits for channel 3 pulse width to be in the range of LOWLIMIT and ‘HIGHLIMIT.  The defined variables must contain the pulse width in microseconds.  

WAIT MP3IDLE
‘Waits for MP3 player to be finished with sound or song.

Appendix A: Example JawaScript Startup.txt file

PLAY 1, “STARTSOUNDS.MP3”
‘ Play start sounds while initializing

SET PSI 1 MODE 1

‘Sets PSI 1 to normal blink mode

SET PSI 2 MODE 1

‘Sets PSI 2 to normal blink mode

SET LOWLIMIT 3000
   
‘First color time in milliseconds

SET HIGHLIMIT 1000

‘Second color time in milliseconds

SET PSI 1 RATE
       ‘Set Front PSI initial times (msec)  

SET PSI 2 RATE  
       ‘Set Rear PSI initial times (msec)

SET PSI 1 TRANS 250        ‘Set PSI transition time (msec)

SET PSI 2 TRANS 250        ‘Set PSI transition time (msec)

SET FLD 1 MODE 1
      
‘Set Top FLD to random pattern

SET FLD 2 MODE 1          
‘Set Bottom FLD to random pattern

SET CHANNEL 1, 1520        ‘Set Initial channel 1 center value

SET CHANNEL 2, 1520        ‘Set Initial channel 2 center value

SET CHANNEL 3, 1520        ‘Set Initial channel 3 center value

SET CHANNEL 4, 1520        ‘Set Initial channel 4 center value

SET CHANNEL 5, 1520        ‘Set Initial channel 5 center value

SET CHANNEL 6, 1520        ‘Set Initial channel 6 center value

SET SERVO 1 CENTER 1520 
‘Set servo 1 center to 1520 microseconds  

SET SERVO 2 CENTER 1520 
‘Set servo 2 center to 1520 microseconds  

SET SERVO 3 CENTER 1520 
‘Set servo 3 center to 1520 microseconds  

SET SERVO 1 DEADBAND 20    ‘Set servo 1 deadband to 20 microseconds  

SET SERVO 2 DEADBAND 20    ‘Set servo 2 deadband to 20 microseconds  

SET SERVO 3 DEADBAND 20    ‘Set servo 3 deadband to 20 microseconds  

ASSIGN SERVO 1, 1        ‘Assign Servo 1 to Channel 1 (Right Drive motor controller)

ASSIGN SERVO 2, 2 INVERSE ‘Assign Servo 2 to Channel 2 inverse (Left Drive motor

‘controller)

ASSIGN SERVO 3, 4   
‘Assign Servo 3 to Channel 4 (Dome motor controller)

SET JEDI 2, 14789 “Leia.txt”
‘Set stroke “L” to play Leia message.

WAIT MP3IDLE


‘Wait for MP3 sounds to finish

PLAY 1,”FINISHSOUND.MP3” 
‘Play finish sound

Appendix B: Example JawaScript Fire Extinguisher file

PLAY 1, “ALARM.MP3”

‘Play the Alarm sound

SET PSI 1 MODE 1        
‘Set Front PSI to normal display mode 

SET LOWLIMIT 100

‘Setup variable for use

SET HIGHLIMIT 100

‘Setup variable for use

SET PSI 1 RATE        
‘Set Front PSI to 200 msec rate, even red and blue times 

SET FLD 1 MODE 3          
‘Sets Top FLD to alarm pattern

SET FLD 2 MODE 3          
‘Sets Bottom FLD to alarm pattern

PARK 6             

‘Park the current channel 6 assignments

SET SERVO 12, 1000        
‘Set servo 12 to +45 degrees (open top tool arm) 

WAIT 1000


‘Wait 1000 msec (1 sec) for tool arm to open

ASSIGN HCOUT 3, 6, INVERSE ‘Channel 6 to extinguisher solenoid, use down button

WAIT JEDI         
    
‘Wait for new J.E.D.I. or Null command


ASSIGN HCOUT 3, OFF       
‘High Current Output 3 back to default value

SET SERVO 12, 1520
     
‘Reset servo 12 to 0 degrees (close top arm)

UNPARK 6                  
‘Restore channel 6 assignment to last assignment

RESET PSI 1               
‘Set Front PSI to normal rate 

RESET FLD 1               
‘Sets Top FLD to normal pattern

RESET FLD 2               
‘Sets Bottom FLD to normal pattern

Appendix C: Wish List

This is a wish list of JawaScript commands.  I doubt they will ever be implemented due to lack of time and memory resources, but listed just in case….  Never say never! 
In priority order:
1. User rename of objects and constants for clarity (RENAME command)

2. Some user variables… would be nice. Probably pre-defined as VAR1 to VARn

3. LOOP and END LOOP commands…. No CLI support, of course.
4. Conditional execution… Probably no CLI support, haven’t found need.

5. Logging commands for debug would be nice.

Legal Stuff

All artwork, block diagrams, non-open source software, scripting language, and this documentation are copyright by Scott Gray, 2007.  Replication or reproduction in whole or in part is strictly prohibited without express written consent from Scott Gray. 

Software specifications can and will change without notice and I will not be liable for any changes.  Assume there are syntax errors in this document.  Use the examples herein as guidelines; the syntax is defined by the command syntax, unless there is a syntactical error in that syntax.  In that case, the examples may have the correct syntax.  Proceed with caution (Trust, but verify).

